# Maintainer: Andreas Radke <andyrtr@archlinux.org>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xmessage
pkgver=1.0.6
pkgrel=1
pkgdesc="Display a message or query in a window"
arch=(loong64 x86_64)
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxaw')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xmessage-${pkgver}.tar.xz{,.sig})
sha512sums=('89deb7a81bd047027f99cb92cd16a47f94479c61a6f9fbac4638e6d642037b3e9bbac2313db59ac8c5544175947c693f6a210537c6df2b88dc3de1c67498cba8'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd xmessage-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xmessage-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
